


### Harvesting a Population

The problem that we are trying to solve is how to use an ODE to determine harvesting of a population; more specifically deriving and understanding the model of harvesting using a chemostat. The most basic population is a model in which exponential growth is achieved.

$$P' = kP - H$$

This equation, however useful, can only be applied in models where there is a constant rate of harvesting. There are some situations where we need a more complex model. One of these is the model for bacteria growth in a chemostat.

Before we can go into the complex derivation of a model for population in a chemostat, we must first understand what one is. A chemostat is an apparatus used to harvest microorganisms. It has two main parts: a nutrient reservoir and a growth chamber. The nutrients, needed for cell growth, are continuously pumped from the reservoir into the growth chamber. Residual cells and nutrients flow out of the growth chamber at the same rate that nutrients flow in, maintaining a constant volume. This makes a chemostat very useful. It has commercial applications in the manufacturing of biological material such as insulin. It is also used in the laboratory to study the growth and competition of microorganisms. A chemostat is preferable over other techniques



because parameters are kept constant, and thus the results are accurate and reproducible. Also, the chemostat allows scientists to study the properties of organisms at chosen growth rates. This is useful because microorganisms rarely grow at their maximum rate in their natural environment.

To determine the harvest rate of chemostat we will use two equations; one for the bacteria growth in Chemostat and one for the nutrients.

```
> Bacteria := diff(B(t), t) = beta*(N(t)/(1+N(t)))*B(t) - B(t);
```

$$\text{Bacteria} := \frac{d}{dt}B(t) = \frac{\beta N(t) B(t)}{1 + N(t)} - B(t)$$

> **Nutrient := diff(N(t), t) = -(N(t)/(1+N(t)))\*B(t) - N(t) + gamma;**

$$\text{Nutrient} := \frac{d}{dt}N(t) = -\frac{N(t) B(t)}{1 + N(t)} - N(t) + \gamma$$

There are two different scenarios in which  $dB/dt = 0$ , also known as nullclines. Nullclines are steady states of systems of equations which can be used to sketch a solution. To find these nullclines, the right hand side of  $dB/dt$  can be set equal to zero and solved for either  $B(t)$  or  $N(t)$ :

> **B\_Nullcline\_Beqn:=solve(rhs(Bacteria)=0,B(t));**

$$B\_Nullcline\_Beqn := 0$$

This first solution is the value of  $B$  in which  $dB/dt = 0$ .

> **N\_Nullcline\_Beqn:=solve(rhs(Bacteria)=0,N(t));**

$$N\_Nullcline\_Beqn := \frac{1}{\beta - 1}$$

The second solution is the value of  $N$  for which  $dB/dt = 0$ . Notice that this solution is a line parallel to the  $B$  axis, because the value of  $N$  does not depend upon  $B$ .

There are again two scenarios in which  $dN/dt = 0$ . One solution is the value of  $B$  for which  $dN/dt = 0$ :

> **B\_Nullcline\_Neqn:=solve(rhs(Nutrient)=0,B(t));**

$$B\_Nullcline\_Neqn := -\frac{-\gamma N(t) + N(t) + N(t)^2 - \gamma}{N(t)}$$

This equation can then be factored into:

> **B\_Nullcline\_Neqn := factor(B\_Nullcline\_Neqn);**

$$B\_Nullcline\_Neqn := -\frac{(1 + N(t))(N(t) - \gamma)}{N(t)}$$

The solution is a curve with a value of  $B = 0$  when  $N = \gamma$ .

The chemostat model is made up of two equations in two variables, thus the behavior of the model can be graphed. The axes for this graph are  $B$  and  $N$ ; the variable put on each axis is an arbitrary decision. In order to plot the solution of the  $B$  and  $N$  equations we

need to define the solutions as functions of B. Therefore,  $N=f(B)$  for the  $dB/dt=0$  equation and  $N=g(B)$  for the  $dN/dt=0$  equation.

> **f:=N\_Nullcline\_Beqn;**

$$f := \frac{1}{\beta - 1}$$

> **B=B\_Nullcline\_Neqn;**

$$B = -\frac{(1 + N(t))(N(t) - \gamma)}{N(t)}$$

In order to make the equation a function we can graph we move N to the left hand side and use a different variable for the right. Then we solve the equation for N.

> **g:=solve(B(t)=B\_Nullcline\_Neqn,N(t));**

$$g := -\frac{1}{2}B(t) - \frac{1}{2} + \frac{1}{2}\gamma + \frac{1}{2}\sqrt{B(t)^2 + 2B(t) - 2B(t)\gamma + 1 + 2\gamma + \gamma^2}, -\frac{1}{2}B(t) - \frac{1}{2} + \frac{1}{2}\gamma - \frac{1}{2}\sqrt{B(t)^2 + 2B(t) - 2B(t)\gamma + 1 + 2\gamma + \gamma^2}$$

The solution is messy and is actually two separate solutions. The solutions can be seen when the solution is plotted on the axes N and B. In order to plot we will define the solutions. The first solution is g1(N) and the second it g2(N).

> **g1:=op(1, [g]);**

$$g1 := -\frac{1}{2}B(t) - \frac{1}{2} + \frac{1}{2}g + \frac{1}{2}\sqrt{B(t)^2 + 2B(t) - 2B(t)g + 1 + 2g + g^2}$$

> **g2:=op(2, [g]);**

$$g2 := -\frac{1}{2}B(t) - \frac{1}{2} + \frac{1}{2}g - \frac{1}{2}\sqrt{B(t)^2 + 2B(t) - 2B(t)g + 1 + 2g + g^2}$$

To plot a graph of the model, values of beta and gamma need to be chosen. Beta, the bacteria's growth rate, must be greater than 1 so that the equilibrium nutrient concentration is not zero or less than zero. Gamma, the input flow rate, must be greater than zero because in a chemostat, there is a constant flow into the growth chamber. The following statement defines beta and gamma. The starting parameter values are beta=4 and gamma=1


> **parameters:=beta=4, gamma=1:**

> **G:=subs(parameters,g1):**

> **F:=subs(parameters,f):**


> **N\_nullcline:=plot(G,B=0..20,color=red):**

```
> B_nullcline:=plot(F,B=0..20,color=blue):
> display(N_nullcline,B_nullcline);
```



$g(B)$ , an alternate function, is  $g_2$ , and its graph is:

```
> G:=subs(parameters,g2):
> N_nullcline1:=plot(G,B=0..20,color=green):
> display(N_nullcline1,B_nullcline);
```



The only possible solution that correctly models a chemostat has positive nutrient concentration values and positive population values.

If we change the values of the parameter  $\beta$ , we must keep a few things in mind. Since the nutrient concentration,  $N$ , is equal to the inverse of  $\beta-1$ , it is necessary that  $B$  be greater than 1, otherwise  $N$  would be infinite ( $\beta=1$ ) or negative ( $\beta < 1$ ).


Mathematician, Edelstein-Keshet, notes another constraint on the model. She claims that  $\gamma$  must be greater than  $\frac{1}{\beta-1}$ . Let's try it when this condition is not met; we'll set  $\beta = 1.5$  and  $\gamma = 1$ .

```
> parameters:=beta=1.5, gamma=1:
> G:=subs(parameters,g1):
```

```

> F:=subs(parameters,f):
> N_nullcline:=plot(G,B=0..20,color=red):
> B_nullcline:=plot(F,B=0..20,color=blue):
> display(N_nullcline,B_nullcline);

```




The alternate function  $g(B)$  is  $g_2$ , and its graph is

```

> G:=subs(parameters,g2):
> N_nullcline1:=plot(G,B=0..20,color=green):
> display(N_nullcline1,B_nullcline);

```



As you can see there is no point where the nullclines will meet in the realm of positive numbers.

We will now try looking at the dynamics of the chemostat system. In order to do this, we define the equations, the variables, the time domain, the initial conditions, and tell the system to plot the lines in particular colors. This is done with the DEtools package and the DEplot command.


```

> equations:=subs(parameters,[Bacteria,Nutrient]):
> vars:=[B(t),N(t)]:
> domain:=t=0..20:
> IC:=[[0,1,0],[0,3,1],[0,0.01,0.01],[0,5,.01]]:

```


```
> dynamics := DEplot(equations,vars,domain,IC,scene=[B,N], linecolor=[red,green,blue,black], stepsize=0.1):
```

```
> display(dynamics);
```



We can now display the dynamics on top of the nullcline graphs

```
> display(dynamics,B_nullcline,N_nullcline);
```




As you can tell from this graph, the place where both the nullclines are equal is the location where all the dynamic graphs converge to.

```
> N_dynamics :=DEplot(equations,vars,domain,IC,scene=[t,N], linecolor=[red,green,blue,black], stepsize=0.1):
```

```
> B_dynamics :=DEplot(equations,vars,domain,IC,scene=[t,B], linecolor=[pink,aquamarine,navy,gre], stepsize=0.1):
```


```
> display(N_dynamics,B_dynamics,title="Chemostat Dynamics - (N dark, B pale)");
```



As you can see from this graph, no matter the initial data we are given, the nutrient concentration (dark lines) and the bacteria concentration (pale lines) will converge to the same values every time. These values are characteristic to each chemostat and its  $\beta$  and  $\gamma$  values.

If  $\gamma$ , the flow rate of the system, were to be changed while keeping all other variables constant, the amount of bacteria produced would increase. This is due to an increase in nutrients entering the system. With more nutrients available, bacteria are able to grow at a faster rate. For example, if  $\gamma$  were taken to be 5 instead of 1 in the examples above:

```
> parameters2 := beta=4, gamma=5:
> equations2 := subs(parameters2, [Bacteria, Nutrient]):
> dynamics2 := DEplot(equations2, vars, domain, IC, scene=
[B, N], linecolor=[red, green, blue, black], stepsize=0.1):
> display(dynamics2);
```



As is shown in the graph, after  $t=20$ , the graph spirals inward towards a  $(B,N)$  value of approximately  $(18,0.35)$  with a  $\gamma$  value of 5 as opposed to approximately  $(2.6,0.35)$  when  $\gamma$  is equal to 1.

## Bibliography

Davis, Paul W. Differential Equations: Modeling with Matlab. Upper Saddle River, NJ: Prentice Hall, 1999.

Edelstein-Keshet, Leah. Mathematical Models in Biology. New York, NY: Random House, 1988.

Miller, Matthew. Graphical Analysis of the chemostat equations.

[http://www.maplesoft.com/applications/app\\_center\\_view.aspx?AID=450](http://www.maplesoft.com/applications/app_center_view.aspx?AID=450)

Milsom, William. "Chemostat." McGraw-Hill Encyclopedia of Science & Technology. 2002 ed.

Pun, Pattle P. T. Evolution: Nature and Scripture in Conflict?

[http://www.ibri.org/Books/Pun\\_Evolution/Chapter3/3.2.htm](http://www.ibri.org/Books/Pun_Evolution/Chapter3/3.2.htm)